

# Count cells between dates

**Count cells between dates**

Name	Birthday
Jim	1/15/1999
Aya	2/5/1999
Sally	3/23/2000
Charles	5/1/2000
Terry	5/11/2000
Joe	8/20/2000
Sarah	1/3/2001

Year	Birthdays
1999	2
2000	4
2001	1
2002	0

*dates = C5:C11*

**EXCELJET**

## Generic formula

```
= COUNTIFS (range, ">=" & date1, range, "<=" & date2)
```

## Summary

To count the number of cells that contain dates between two dates, you can use the [COUNTIFS function](#). In the example shown, F6 contains this formula:

```
= COUNTIFS (dates, ">=" & DATE (E5, 1, 1), dates, "<=" & DATE (E5, 12, 31))
```

This formula counts birthdays in the year 2000, that appear in the range C5:C11.

*Note: this formula uses the [named range](#) "dates", C5:C11.*

## Explanation

The COUNTIFS function is built to count cells that meet multiple criteria. In this case, we need to provide two criteria: one criteria for the earlier date and one for the later date. We supply the [named range](#) **dates** (C5:C11) for both criteria.

To construct each date, we use the [DATE function](#):

```
DATE (E5, 1, 1) // build first day of year  
DATE (E5, 12, 31) // build last day of year
```

The DATE function makes it easy to build dates based on year, month, and day arguments that are either hard-coded or supplied as cell references. In the example, month and day are hard-coded, and we get year from column E.

Note: the [operators](#) ">=" and "<=" must be entered as text and surrounded by double quotes. This means we must use [concatenation](#) (&) to join each operator to each date.